

Calculus Single Variable

Brian E. Blank and Steven G. Krantz

Section 6.6 Inverse Trigonometric Functions

1. The Function Sin and its Inverse, arcsin

Define the function $x \rightarrow \text{Sin}(x)$ by $\text{Sin}(x) = \sin(x)$ for $x = -\frac{\pi}{2} \dots \frac{\pi}{2}$. In other words, we define

Sin by restricting the domain of sine to $-\frac{\pi}{2} \dots \frac{\pi}{2}$. Because the domain of Sin , namely $-\frac{\pi}{2} \dots \frac{\pi}{2}$, is not the same as the domain of sine, namely $-\infty \dots \infty$, these two functions are different. The use of a new name, Sin , is therefore appropriate.

Observe that Sin is an increasing, hence invertible, function with domain $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$ and image $[-1, 1]$.




Figure 1

The inverse of Sin is denoted arcsin . The domain of arcsin is $[-1, 1]$ and its image is $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$. We have

$$\begin{aligned} \text{arcsin}(\sin(x)) &= x && \text{for } x \text{ in } \left[-\frac{\pi}{2}, \frac{\pi}{2}\right], \text{ and} \\ \sin(\text{arcsin}(y)) &= y && \text{for } y \text{ in } [-1, 1]. \end{aligned}$$

The plot of $y = \text{arcsin}(x)$ is given in Figure 2.




Figure 2

Exercise:

Calculate $\text{arcsin}\left(\frac{1}{2}\right)$ and $\text{arcsin}\left(\sin\left(\frac{3\pi}{2}\right)\right)$.

Solution:

There are many values of x for which $\sin(x) = \frac{1}{2}$ but only one value, $x = \frac{\pi}{6}$, in the interval $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$. Thus, $\text{arcsin}\left(\frac{1}{2}\right) = \frac{\pi}{6}$.

Similarly, there are many values of x for which $\sin(x) = -1$ but only one value, $x = -\frac{\pi}{2}$, is in the interval $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$.

Thus, $\arcsin\left(\sin\left(\frac{3\pi}{2}\right)\right) = \arcsin(-1) = -\frac{\pi}{2}$. Notice that the answer is not $-\frac{3\pi}{2}$. In particular, the equation $\arcsin(\sin(x)) = x$ is false as a general rule. It is true only for x in the interval $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$.

- 2. The Derivative of arcsin

For x in the interval $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$ we have

$$\arcsin(\sin(x)) = x.$$

Therefore

$$\left[\begin{array}{l} > \text{Diff}(\arcsin(\sin(x)), x) = \text{Diff}(x, x); \\ \frac{d}{dx} \arcsin(\sin(x)) = \frac{d}{dx} x \end{array} \right.$$

The right side is 1 and, from an application of the Chain Rule, the left side becomes $D(\arcsin)(\sin(x)) \left(\frac{d}{dx} \sin(x)\right)$. Thus,

$$\left[\begin{array}{l} > 'D(\arcsin)(\sin(x))' * \text{diff}(\sin(x), x) = 1; \\ D(\arcsin)(\sin(x)) \cos(x) = 1 \end{array} \right.$$

or

$$\left[\begin{array}{l} > 'D(\arcsin)(\sin(x))' = 1/\cos(x); \\ D(\arcsin)(\sin(x)) = \frac{1}{\cos(x)} \end{array} \right.$$

Let $y = \sin(x)$. Then $\cos(x)^2 = 1 - y^2$. Notice that $0 \leq \cos(x)$ because x is in the interval $\left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$. Therefore we use the positive root when we solve for $\cos(x)$: $\cos(x) = \sqrt{1 - y^2}$. This results in the differentiation formula

$$D(\arcsin)(y) = \frac{1}{\sqrt{1 - y^2}}.$$

Notice that the derivative of arcsin is always positive. That is because Sin is an increasing function. The inverse of an increasing function is also increasing, and therefore has a positive derivative.

Exercise:

Let a be a positive constant. Calculate the derivative of $x \rightarrow \arcsin\left(\frac{x}{a}\right)$.

Solution:

By the Chain Rule

```
> eqn := Diff(arcsin(x/a), x) = subs(u = x/a,
  D(arcsin)(u))*Diff(x/a, x);
```

$$eqn := \frac{\partial}{\partial x} \arcsin\left(\frac{x}{a}\right) = \frac{\frac{\partial}{\partial x} \left(\frac{x}{a}\right)}{\sqrt{1 - \frac{x^2}{a^2}}}$$

We have only to tidy up:

```
> lhs(eqn) = simplify(value(rhs(eqn))) assuming a > 0;
```

$$\frac{\partial}{\partial x} \arcsin\left(\frac{x}{a}\right) = \frac{1}{\sqrt{a^2 - x^2}}$$

This calculation yields a useful integral formula:

$$\int \frac{1}{\sqrt{a^2 - x^2}} dx = \arcsin\left(\frac{x}{a}\right) + C .$$

3. The Function Cos and its Inverse, arccos

Define the function $x \rightarrow \text{Cos}(x)$ by $\text{Cos}(x) = \cos(x)$ for $x = 0 \dots \pi$. In other words, we define Cos by restricting the domain of cosine to $0 \dots \pi$. Because the domain of Cos , namely $0 \dots \pi$, is not the same as the domain of cosine, namely $-\infty \dots \infty$, these two functions are different. The use of a new name, Cos , is therefore appropriate.

Observe that Cos is a decreasing, hence invertible, function with domain $[0, \pi]$ and image $[-1, 1]$.

The inverse of Cos is denoted arccos . The domain of arccos is $[-1, 1]$ and its image is $[0, \pi]$. We have

$$\begin{aligned} \text{arccos}(\cos(x)) &= x && \text{for } x \text{ in } [0, \pi], \text{ and} \\ \cos(\text{arccos}(y)) &= y && \text{for } y \text{ in } [-1, 1]. \end{aligned}$$

The plot of $y = \text{arccos}(x)$ is given in Figure 3.




Figure 3

4. The Function Tan and its Inverse, arctan

Define the function $x \rightarrow \text{Tan}(x)$ by $\text{Tan}(x) = \tan(x)$ for $-\frac{\pi}{2} < x < \frac{\pi}{2}$. In other words, we define Tan by restricting the domain of \tan to the open interval from $-\frac{\pi}{2}$ to $\frac{\pi}{2}$.

Observe that Tan is an increasing, hence invertible, function. The inverse of Tan is denoted arctan .

$$\text{arctan}(\tan(x)) = x \quad \text{for } -\frac{\pi}{2} < x < \frac{\pi}{2}, \text{ and}$$

$$\tan(\text{arctan}(y)) = y \quad \text{for } -\infty < y < \infty.$$

The plot of $y = \text{arctan}(x)$ is given in Figure 4.




Figure 4

For $-\frac{\pi}{2} < x < \frac{\pi}{2}$ we have

$$\arctan(\tan(x)) = x.$$

Therefore

$$\left[\begin{array}{l} > \text{Diff}(\arctan(\tan(x)), x) = \text{Diff}(x, x); \\ \qquad \qquad \qquad \frac{d}{dx} \arctan(\tan(x)) = \frac{d}{dx} x \end{array} \right.$$

The right side is 1 and, from an application of the Chain Rule, the left side becomes

$$D(\arctan)(\tan(x)) \left(\frac{d}{dx} \tan(x) \right). \text{ Thus,}$$

$$\left[\begin{array}{l} > 'D(\arctan)(\tan(x))' * \sec(x)^2 = 1; \\ \qquad \qquad \qquad D(\arctan)(\tan(x)) \sec(x)^2 = 1 \end{array} \right.$$

or

$$\left[\begin{array}{l} > 'D(\arctan)(\tan(x))' = 1/\sec(x)^2; \\ \qquad \qquad \qquad D(\arctan)(\tan(x)) = \frac{1}{\sec(x)^2} \end{array} \right.$$

Let $y = \tan(x)$. Then $\sec(x)^2 = 1 + \tan(x)^2 = 1 + y^2$. These substitutions result in the differentiation formula

$$D(\arctan)(y) = \frac{1}{1 + y^2} \bullet$$

Notice that the derivative of arctan is always positive. That is because Tan is an increasing function. The inverse of an increasing function is also increasing, and therefore has a positive derivative.

Exercise:

Let a be a positive constant. Calculate the derivative of $x \rightarrow \arctan\left(\frac{x}{a}\right)$.

Solution:

By the Chain Rule

```
> eqn := Diff(arctan(x/a), x) = subs(u = x/a,  
  D(arctan)(u))*Diff(x/a, x);
```

$$eqn := \frac{\partial}{\partial x} \arctan\left(\frac{x}{a}\right) = \frac{\frac{\partial}{\partial x} \left(\frac{x}{a}\right)}{1 + \frac{x^2}{a^2}}$$

We have only to tidy up:

```
> lhs(eqn) = simplify(value(rhs(eqn))) assuming a > 0;
```

$$\frac{\partial}{\partial x} \arctan\left(\frac{x}{a}\right) = \frac{a}{a^2 + x^2}$$

This calculation yields a useful integral formula:

$$\int \frac{1}{a^2 + x^2} dx = \frac{1}{a} \arctan\left(\frac{x}{a}\right) + C .$$

5. The Function Sec and its Inverse, arcsec

Define the function $x \rightarrow \text{Sec}(x)$ by $\text{Sec}(x) = \sec(x)$ for $0 \leq x < \frac{\pi}{2}$ and $\frac{\pi}{2} < x \leq \pi$. In other words, we define Sec by restricting the domain of sec. The point $\frac{\pi}{2}$ is excluded from the domain of Sec because it is not in the domain of sec.




Figure 5




Figure 6

Observe that Sec is an increasing, hence invertible, function.
 The inverse of Sec is denoted arcsec .

$$\text{arcsec}(\sec(x)) = x \quad \text{for } 0 \leq x < \frac{\pi}{2} \text{ and } \frac{\pi}{2} < x \leq \pi, \text{ and}$$

$$\sec(\text{arcsec}(y)) = y \quad \text{for } 1 \leq |y|.$$

The plot of $y = \text{arcsec}(x)$ is given in Figure 7.




Figure 7

For $0 \leq x < \frac{\pi}{2}$ and $\frac{\pi}{2} < x \leq \pi$ we have

$$\operatorname{arcsec}(\sec(x)) = x.$$

Therefore

$$\left[\begin{array}{l} > \text{Diff}(\operatorname{arcsec}(\sec(x)), x) = \text{Diff}(x, x); \\ \qquad \qquad \qquad \frac{d}{dx} \operatorname{arcsec}(\sec(x)) = \frac{d}{dx} x \end{array} \right.$$

The right side is 1 and, from an application of the Chain Rule, the left side becomes

$$D(\operatorname{arcsec})(\sec(x)) \left(\frac{d}{dx} \sec(x) \right). \text{ Thus,}$$

$$\left[\begin{array}{l} > 'D(\operatorname{arcsec})(\sec(x))' * \text{diff}(\sec(x), x) = 1; \\ \qquad \qquad \qquad D(\operatorname{arcsec})(\sec(x)) \sec(x) \tan(x) = 1 \end{array} \right.$$

or

$$\left[\begin{array}{l} > 'D(\operatorname{arcsec})(\sec(x))' = 1/\sec(x)/\tan(x), \quad \text{Preliminary} \\ \text{formula for } D(\operatorname{arcsec})'; \\ \qquad \qquad \qquad D(\operatorname{arcsec})(\sec(x)) = \frac{1}{\sec(x) \tan(x)}, \quad \text{Preliminary formula for } D(\operatorname{arcsec}) \end{array} \right.$$

Let $y = \sec(x)$. Then $\tan(x)^2 = \sec(x)^2 - 1 = y^2 - 1$. It follows that $|\tan(x)| = \sqrt{y^2 - 1}$ and

$$\sec(x) |\tan(x)| = y \sqrt{y^2 - 1}. \quad \text{Formula with sign to be resolved.}$$

There are two cases.

Case 1: If $0 \leq x < \frac{\pi}{2}$, then $0 < y$ and $0 < \tan(x)$. Therefore $y = |y|$ and $|\tan(x)| = \tan(x)$.

The formula with the sign to be resolved can be written as

$$\sec(x) \tan(x) = |y| \sqrt{y^2 - 1}. \quad \text{Formula with sign resolved, } 0 \leq x < \frac{\pi}{2}.$$

Case 2: If $\frac{\pi}{2} < x \leq \pi$, then $y < 0$ and $\tan(x) < 0$. Therefore $y = -|y|$ and $|\tan(x)| = -\tan(x)$.

The formula with the sign to be resolved can be written as

$$\begin{aligned} -\sec(x) \tan(x) &= y \sqrt{y^2 - 1}, \text{ or} \\ \sec(x) \tan(x) &= -y \sqrt{y^2 - 1}, \text{ or} \\ \sec(x) \tan(x) &= |y| \sqrt{y^2 - 1}. \end{aligned}$$

In this second case we also have

$$\sec(x) \tan(x) = |y| \sqrt{y^2 - 1}. \quad \text{Formula with sign resolved, } \frac{\pi}{2} < x \leq \pi.$$

The sign issue has now been resolved with one formula common to both cases. We have obtained the following differentiation formula:

$$D(\operatorname{arcsec})(y) = \frac{1}{|y| \sqrt{y^2 - 1}}.$$

Notice that $0 < D(\operatorname{arcsec})(y)$ for all permissible y , as it must be (since arcsec is an increasing function).

Remark: **Maple** has an interesting way to handle the sign issue without resorting to the use of absolute values:

`> D(arcsec)(y);`

$$\frac{1}{y^2 \sqrt{1 - \frac{1}{y^2}}}$$

Exercise:

Let a be a positive constant. Calculate the derivative of $x \rightarrow \operatorname{arcsec}\left(\frac{x}{a}\right)$.

Solution:

By the Chain Rule

```
> eqn := Diff(arcsec(x/a), x) = subs(u = x/a,
  D(arcsec)(u))*Diff(x/a, x);
```

$$eqn := \frac{\partial}{\partial x} \operatorname{arcsec}\left(\frac{x}{a}\right) = \frac{a^2 \left(\frac{\partial}{\partial x} \left(\frac{x}{a} \right) \right)}{x^2 \sqrt{1 - \frac{a^2}{x^2}}}$$

We have only to tidy up:

```
> lhs(eqn) = simplify(value(rhs(eqn))) assuming a > 0;
```

$$\frac{\partial}{\partial x} \operatorname{arcsec}\left(\frac{x}{a}\right) = \frac{a}{x^2 \sqrt{\frac{x^2 - a^2}{x^2}}}$$

This calculation yields a useful integral formula:

$$\int \frac{1}{|x| \sqrt{x^2 - a^2}} dx = \frac{1}{a} \operatorname{arcsec}\left(\frac{x}{a}\right) + C .$$

Code

The code for Figure 1:

```
> SinePlot := plot(sin(x), x = -Pi/2 .. Pi/2, scaling =
  constrained, thickness = 2, color = PLUM, view =
  [-Pi/2..Pi/2,-Pi/2..Pi/2]):
  legend := plots[textplot]([0.2, 1, `y = Sin(x)`],
  align={ABOVE,RIGHT}, color = PLUM):
  plots[display](SinePlot,legend, tickmarks=[3,3])
```

The code for Figure 2

```
> arcsinPlot := plot(arcsin(x), x = -1 .. 1, scaling =
  constrained, thickness = 2, color = NAVY, view =
  [-Pi/2..Pi/2,-Pi/2..Pi/2]):
  SinePlot := plot(sin(x), x = -Pi/2 .. Pi/2, scaling =
  constrained, thickness = 2, color = PLUM, view =
  [-Pi/2..Pi/2,-Pi/2..Pi/2]):
  linePlot := plot(x, x = -Pi/2..Pi/2, scaling = constrained,
  linestyle = 4, thickness = 1, color = GRAY, view =
  [-Pi/2..Pi/2,-Pi/2..Pi/2]):
  legend1 := plots[textplot]([0.1, 1.4, `y = arcsin(x)`],
  align={BELOW,RIGHT}, color = NAVY):
  legend2 := plots[textplot]([-1.5, -.6, `y = Sin(x)`],
  align={BELOW,RIGHT}, color = PLUM):
  plots[display](arcsinPlot, SinePlot, linePlot, legend1,
  legend2, tickmarks=[3,3]);
```

The code for Figure 3

```
> arccosPlot := plot(arccos(x), x = -1 .. 1, scaling =
  constrained, thickness = 2, color = NAVY, view =
  [-1..Pi,-1..Pi]):
cosPlot := plot(cos(x), x = 0 .. Pi, scaling = constrained,
  thickness = 2, color = PLUM, view = [-1..Pi,-1..Pi]):
linePlot := plot(x, x = -1..Pi, scaling = constrained,
  linestyle = 4, thickness = 1, color = GRAY, view =
  [-1..Pi,-1..Pi]):
legend1 := plots[textplot]([0.25, 1.6, `y = arccos(x)`],
  align={BELOW,RIGHT}, color = NAVY):
legend2 := plots[textplot]([1.25, 0.45, `y = Cos(x)`],
  align={ABOVE,RIGHT}, color = PLUM):
plots[display](arccosPlot, cosPlot, linePlot, legend1, legend2,
  tickmarks=[3,3]);
```

The code for Figure 4

```
> arctanPlot := plot(arctan(x), x = -5 .. 5, scaling =
  constrained, thickness = 2, color = NAVY, view =
  [-5..5,-5..5]):
tanPlot := plot(tan(x), x = -Pi/2 .. Pi/2, scaling =
  constrained, thickness = 2, color = PLUM, view =
  [-5..5,-5..5]):
linePlot := plot(x, x = -5..5, scaling = constrained, linestyle
  = 4, thickness = 1, color = GRAY, view = [-5..5,-5..5]):
legend1 := plots[textplot]([2.75, 1.7, `y = arctan(x)`],
  align={ABOVE,RIGHT}, color = NAVY):
legend2 := plots[textplot]([1.6, 4.3, `y = Tan(x)`],
  align={ABOVE,RIGHT}, color = PLUM):
plots[display](arctanPlot, tanPlot, linePlot, legend1, legend2,
  tickmarks=[3,3]);
```

The code for Figure 5

```
> secPlot := plot(sec(x), x = 0 .. 2*Pi, discontinuous=true, scaling =
constrained, thickness = 2, color = MAGENTA, view =
[0..5,-5..5]):
leftVertLinePlot := plot([Pi/2,y, y = -5..5], scaling =
constrained, linestyle = 4, thickness = 1, color = PLUM, view =
[0..5,-5..5]):
rightVertLinePlot := plot([3*Pi/2,y, y = -5..5], scaling =
constrained, linestyle = 4, thickness = 1, color = PLUM, view =
[0..5,-5..5]):
legend := plots[textplot]([2.2, -3.4, `y = sec(x)`],
align={ABOVE,RIGHT}, color = MAGENTA):
plots[display](secPlot, leftVertLinePlot, rightVertLinePlot,
legend, tickmarks=[3,3]):
```

The code for Figure 6

```
> secPlot := plot(sec(x), x = Pi .. 2*Pi, discontinuous=true, scaling =
constrained, thickness = 2, color = MAGENTA, view =
[0..5,-5..5]):
SecPlot := plot(sec(x), x = 0 .. Pi, discontinuous=true, scaling =
constrained, thickness = 3, color = COLOR(RED,0.6,0.1,0.6),
view = [0..5,-5..5]):
leftVertLinePlot := plot([Pi/2,y, y = -5..5], scaling =
constrained, linestyle = 4, thickness = 1, color = PLUM, view =
[0..5,-5..5]):
rightVertLinePlot := plot([3*Pi/2,y, y = -5..5], scaling =
constrained, linestyle = 4, thickness = 1, color = PLUM, view =
[0..5,-5..5]):
legend := plots[textplot]([2.2, -3.4, `y = Sec(x)`],
align={ABOVE,RIGHT}, color = COLOR(RED,0.6,0.1,0.6)):
ptPlot :=
plot([[0,1], [0,1.05], [0.05,1], [0,0.95], [-0.05,1], [Pi,-1], [Pi,-1
.05], [Pi-0.05,-1], [Pi,-0.95], [Pi+0.05,-1]], style = POINT,
symbol = CIRCLE, color=COLOR(RED,0.6,0.1,0.6)):
plots[display](secPlot, SecPlot, ptPlot, leftVertLinePlot,
rightVertLinePlot, legend, tickmarks=[3,3]):
```

The code for Figure 7

```
> arcsecPlot := plot(arcsec(x), x = -5 .. 5, scaling =
constrained, thickness = 2, color = NAVY, view =
[-5..5,-5..5]):
secPlot := plot(sec(x), x = 0 .. Pi, discontinuous=true, scaling =
constrained, thickness = 2, color = MAGENTA, view =
[-5..5,-5..5]):
obliqueLinePlot := plot(x, x = -5..5, scaling = constrained,
linestyle = 4, thickness = 1, color = GRAY, view =
[-5..5,-5..5]):
horizLinePlot := plot(Pi/2, x = -5..5, scaling = constrained,
linestyle = 4, thickness = 1, color = BLUE, view =
[-5..5,-5..5]):
vertLinePlot := plot([Pi/2,y, y = -5..5], scaling =
constrained, linestyle = 4, thickness = 1, color = PLUM, view =
[-5..5,-5..5]):
legend1 := plots[textplot]([2.75, 1, `y = arcsec(x)`],
align={BELOW,RIGHT}, color = NAVY):
legend2 := plots[textplot]([2.2, -3.4, `y = Sec(x)`],
align={ABOVE,RIGHT}, color = MAGENTA):
plots[display](arcsecPlot, secPlot, obliqueLinePlot,
horizLinePlot, vertLinePlot, legend1, legend2,
tickmarks=[3,3]);
```

- Copyright and Author Information

Worksheet Title: BlankKrantz-06_6-R8.mws A Maple Release 8 worksheet.

Author: Brian E. Blank

Date Created: 26 January 2000

Date Last Revised: 29 August 2007

This document may not be distributed by any medium, including print, disk, and electronic transfer, without prior written permission of the author.

For more information, please contact the author:

Department of Mathematics,
Washington University in St. Louis
St. Louis, MO 63130

Telephone: (314) 935-6763
e-mail: brian@math.wustl.edu

Copyright: © 2000-2007 Brian E. Blank, All Rights Reserved.